

## 第 8 章 8088 微机系统应用举例

本章基于 8088 最小系统，综合运用 8255, 8253, 8259 等接口芯片以及汇编语言技术，设计了定时采样监控系统和直流电机闭环调速系统。

### 8.1 定时采样监控系统

#### 8.1.1 系统功能要求

利用 8088、8255、8253、8259、ADC0809 设计一个定时采样监控系统，实现功能如下：系统对标准的 0-5V 物理信号每间隔 2 秒进行采样一次，数码管显示采样值(0-255 之间)，当数字量超过 100 时，要求实现声光报警。设定提供给 8253 的基准时钟频率为 1.19MHZ，8088 的基准时钟为 4.47MHZ。

#### 8.1.2 系统组成

定时采样监控系统由采样通道、显示以及声光报警三个部分组成，其原理组成框图如图 8.1 所示。由于 8088 属于单 CPU 的芯片，片上不带任何外设，所以要通过总线拓展外围的接口功能芯片，采用 8253 和 8259 共同实现定时中断，ADC0809 对物理量对应的电压信号进行采样转换，8255 实现对显示数码管和声光报警的管理，系统实现的结构示意如图 8.2 所示。




图 8.1 定时采样监控系统框图




图 8.2 定时采样监控系统结构示意图

### 8.1.3 系统各部分实现

#### 1. 数码管动态显示

系统设计采用 PITA-A 实验台上的硬件资源。实验台上数码管为 4 位共阴极数码管，设计中只用三位。PB0-PB7 管理数码管的段码，PA0-PA2 管理位选。由于 8255 的各端口的带载能力有限，无论输出还是输出都需要外接驱动芯片，系统中采用集电极开路 74LS07 作为驱动。如图 8.2 所示。




图 8.3 8255 管理数码原理示意图

8255 片选和各端口地址分配如表 8.1 所示

表 8.1 8255 端口地址

片选信号线	寄存器	地址
IOY3	A 口	60H
	B 口	61H
	C 口	62H
	控制寄存器	63H

8255 初始化，A、B 口均为输出，工作方式 0-基本输入输出方式。

```
MOV AL, 10000000B
```

```
OUT 63H, AL
```

数码管为共阴，位选信号为 0，则该位显示，利用人眼滞留原理，可同时显示多位数字。数码管动态显示程序主要设计步骤：1) 要显示的数据查表得到对应七段码；2) 段码从 8255 的端口输出；3) 对应的位选为低，其他位为高；4) 延时一段时间。5) 下一次循环。

数码管显示程序的主要问题有：1) 硬件线路线路的问题；2) 数据区的七段码表有误；3) 16 进制转换为 BCD 码程序转换程序有误。4) 查表的指针定位错误。

由于采用的是底层动态刷新的方式，占用了 CPU 大量的资源，在监控程序任务较重的系统中，这种方式不能胜任，应采用专门的数码管管理芯片以减轻 CPU 的负担。

## 2. 声光报警

8255 的 PC0、PC1 分别管理声报警和光报警。8255 各端口的驱动能力为 2.5mA，所以声光报警电路加入驱动，电路示意图如图 8.4 所示。




图 8.4 声光报警电路示意图

### 3. 定时中断设计

#### 1) 定时 2S

利用 8253 的分频的作用，将一个高频的基准频率分频后变成低频输出，即将一个短时间变成了一个较长的时间。

系统中提供给 8253 的基准时钟源为 1.19MHz。

$$T=1/1.19M$$

定时 2S 的初值= $2/T=2.38 \times 10^6=2380000$ ，由于 8253 的单通道最大计数次数为 65535，一个计数器不能满足要求，所以采用通道 0 和通道 2 级联方式，连接示意图如图 8.2 所示。

8253 可编程定时/计数器，其片选和计数器的地址分配如表 8.2。

表 8.2 8253 端口地址

片选信号线	寄存器	地址
IOY2	0#计数器	40H
	1#计数器	41H
	2#计数器	42H
	控制寄存器	43H

初始化程序：

```

MOV AL, 00110110B    ;定时器 0 方式 3
OUT 43H, AL
MOV AX, 1000
OUT 40H, AL
MOV AL, AH
OUT 40H, AL
MOV AL, 10110110B    ;定时器 2 方式 3
OUT 43H, AL

```

```

MOV AX, 2380          ;约 2s
OUT 42H, AL
MOV AL, AH
OUT 42H, AL

```

还有一种方式可以实现单通道计数器定时 2s，利用单通道定时一个基本的时间单元触发中断，在中断服务程序中计数。例如在利用计算器 0 定时 10ms，在中断中计数到 200 次就可以实现定时 2s。

## 2) 中断

定时采样的另一个关键在于设计定时中断服务程序。系统中采用外部 8259 中断控制器实现定时中断源的管理。

中断控制器的使用需要做以下工作：(1) 建立中断矢量表；(2) 初始化 ICW；(3) 初始化 OCW (4) 编写中断服务程序。

系统中 8259 的端口地址和中断矢量表如表 8.3 和表 8.4 所示。

表 8.3 8259 初始化命令字 ICW

	ICW1	ICW2	ICW3	ICW4
系统口地址	20H	21H	21H	21H
	13H	08H		0DH

表 8.4 8259 中断矢量

中断序号	0	1	2	3	4	5	6	7
功能调用	08H	09H	0AH	0BH	0CH	0DH	0EH	0FH
矢量地址	20H	24H	28H	2CH	30H	34H	38H	3CH
	 23H	 27H	 2BH	 2FH	 33H	 37H	 3BH	 3FH

## 4. 采样通道

系统采样的是标准的 0-5V 的标准电压信号，所以采样通道的关键部分在 ADC 芯片的接口设计和应用。本系统通过系统总线拓展 8 位 8 通道的 ADC0809，使用 IN0 采集物理量。ADC 与 CPU 连接如图 8.5 所示，端口地址分配如表 8.5 所示。

根据 ADC0809 的工作时序图，可知启动 AD 转换通过写时序来完成，读取 AD 转换结果的方法可以是延时等待、EOC 中断和 EOC 状态查询三种方法，由于 ADC 转换时间是几十 us，系统中采用延时等待的方法，具体 ADC 控制程序如下。

```

OUT 00H, AL    ;通过写时序启动 AD
CALL DELAY     ;延时等待转换结束
IN AL, 00H     ;读 AD 值，存取 AD 值到 AL 中

```

注：延时应大于 AD 采样转换时间，否则会出现 AD 值不稳定。




图 8.5 A/D 接口连接示意图

表 8.5 ADC0809 的端口地址

地址			选中通道
C	B	A	
0	0	0	IN0
0	0	1	IN1
0	1	0	IN2
0	1	1	IN3
1	0	0	IN4
1	0	1	IN5
1	1	0	IN6
1	1	1	IN7

#### 8.1.4 系统程序

##### 1. 软件设计

系统程序包括：初始化，AD 转换程序、报警处理、16 进制转换 BCD、数码管显示程序以及中断服务程序，主程序和子程序流程图如图 8.6 所示。

汇编语言程序设计中应注意：

- 1) 由于汇编语言的语句有严格的要求，基本上每条语句都要用到固定的寄存器，所以在程序中尽量在数据区定义全局变量，不要采用寄存器保存全局值；
- 2) 变量的名称尽量根据完成的功能和实际意义来定义，在程序中多加注释，以便提高程序的可读性；
- 3) 功能模块尽量写成子程序，既能提高通用性又能增加程序的可读性。子程序调用的参

数入口和结果输出接口尽量简洁明了。如十六进制转换为 BCD 码子程序 CHANGE，入口参数 LEA DI, AD，即把 AD\_DATA 的地址存入 DI。输出结果在变量中 GEI, SHI, BAI。

4) 定时中断服务程序中尽量少做操作和运算，尽量把运算和控制任务放在主程序中。中断服务程序的时间不能大于定时的采样的间隔的时间。

5) 子函数和中断服务程序尽量将用到寄存器保护，压栈程序肯定不会出现错误，不压栈可能出错，注意压栈与出栈的顺序正好相反。




图 8.6 主程序及定时中断服务程序示意图

## 2. 系统参考程序

DATA SEGMENT

NUM\_SEG DB 3FH, 06H, 5BH, 4FH, 06H, 6DH, 7DH, 07H, 7FH, 6FH ;段选

AD\_DATA DB 0 ; 保存 AD 值

FLAG DB 0 ;采样时间到标志位

GEI DB 0 ;十六进制转十进制子程序输出个位，汉语拼音表示，



```

;GE 是关键字，汇编会出现报错，所以用 GEI 表示
SHI DB 0 ;十位
BAI DB 0 ;百位
DATA ENDS
STACK SEGMENT
TOP DB 100 DUP(?)
STACK ENDS
;
;
;
CODE SEGMENT
    ASSUME CS:CODE, DS:DATA, SS:STACK
GO:MOV AX, DATA
    MOV DS, AX
    MOV AX, STACK
    MOV SS, AX
    MOV SP, SIZE TOP
    CALL DELAY ;延时一段时间，等待外部芯片复位就绪
    ;ICW 初始化
    MOV AL, 13H; ICW1
    OUT 20H, AL;
    MOV AL, 08H; ICW2
    OUT 21H, AL;
    MOV AL, 01H; ICW4
    OUT 21H, AL

;中断向量表的建立开始
    CLI
    PUSH DS
    MOV AX, 0000H
    MOV DS, AX
    MOV AX, OFFSET IRQ7
    MOV DS:[003CH], AX
    MOV AX, CS
    MOV DS:[003EH], AX
    POP DS
;开中断
    STI
    IN AL, 21H ;读 8259 中断屏蔽
    AND AL, 7FH ;开 8259 中断 7
    OUT 21H, AL
; 8253 初始化开始
    MOV AL, 00110110B ; 计数器 0 方式 3
    OUT 43H, AL
    MOV AX, 10000 ;计数器初值应写入相应通道
    OUT 40H, AL

```

```

MOV AL, AH
OUT 40H, AL
MOV AL, 10110110B      ; 计数器 2 方式 3
OUT 43H, AL
MOV AX, 2380            ;约 2S
OUT 42H, AL
MOV AL, AH
OUT 42H, AL
; 8255 初始化开始
MOV AL, 10000000B
OUT 63H, AL
MAIN:                   ;主监控程序程序循环体开始
    CMP FLAG, 01H      ;判别标志位
    JNZ DISP           ;不断显示, 等待两秒后中断
    MOV FLAG, 0        ;清零
    OUT 00H, AL        ;启动 AD
    CALL DELAY
    IN AL, 00H         ; 读 AD 采样值
    MOV AD_DATA, AL    ;保存 AD 值
    CMP AD_DATA, 100   ;报警判断
    JAE ALARM
    MOV AL, 00H        ; PC0 输出低电平, PC1 输出低电平, 声光报警关
    OUT 63H, AL
    MOV AL, 02H
    OUT 63H, AL
    JMP NEXT
ALARM:
    MOV AL, 01H        ; PC0 输出高电平, PC1 输出高电平, 声光报警
    OUT 63H, AL
    MOV AL, 03H
    OUT 63H, AL
NEXT:
    LEA DI, AD_DATA    ;十六进制转换成十进制, 个位、十位、百位
    CALL CHANGE        ;十六进制转换成十进制, 个位、十位、百位
DISP:CALL SHOW         ;数码管显示
    JMP MAIN
SHOW PROC NEAR         ;数码管显示子程序
    PUSH AX
    PUSH BX
    LEA BX, NUM_SEG
    MOV AL, GEI        ;显示个位
    XLAT
    OUT 61H, AL        ;段码 B 口输出
    MOV AL, OFEH       ;位选, 个位

```

```

OUT 60H, AL
CALL DELAY1
MOV AL, SHI           ;显示十位
XLAT
OUT 61H, AL           ;段码输出
MOV AL, OFDH          ;位选，十位
OUT 60H, AL
CALL DELAY1
MOV AL, BAI           ;显示百位
XLAT
OUT 61H, AL           ;段码输出
MOV AL, OFBH          ;位选，百位
OUT 60H, AL
CALL DELAY1
POP BX
POP AX
RET
SHOW ENDP
DELAY: PUSH CX         ;延时函数，用于 AD 采样延时
      PUSH AX
      MOV CX, 0040H
BACK: MOV AX, 050CH
ZERO: DEC AX
      JNZ ZERO
      LOOP BACK
      POP AX
      POP CX
      RET
DELAY1: PUSH CX        ;延时函数，用于数码管显示延时
      PUSH AX
      MOV CX, 001H
BACK1: MOV AX, 0100H
ZERO1: DEC AX
      JNZ ZERO1
      LOOP BACK1
      POP AX
      POP CX
      RET
CHANGE PROC NEAR       ;字节十六进制转成十进制，调用前提：LEA DI, XXX
      PUSH AX
      PUSH BX
      PUSH CX
      PUSH DX
      MOV AL, DS: [DI]

```

```

MOV AH, 0
MOV BL, 0AH
DIV BL
MOV GEI, AH
MOV AH, 0
DIV BL
MOV SHI, AH
MOV BAI, AL
POP DX
POP CX
POP BX
POP AX
RET
CHANGE ENDP

```

IRQ7: ;中断服务程序

```

PUSH AX
MOV FLAG, 1 ;FLAG 置 1
MOV AL, 20H
OUT 20H, AL ;发 EOI 命令
POP AX
IRET
CODE ENDS
END GO

```

程序可以将 AD 采样部分直接放在中断服务程序中，这样采样时间更精准。